

TF4062: Implementing Density Functional Theory using Finite Difference Method

Fadjar Fathurrahman
Iwan Prasetyo

1 Introduction

Density functional theory: [1, 2],

Applications: [3]

Books: [4, 5, 6]

Using the Kohn-Sham density functional theory, total energy of system of interacting electrons under external potential $V_{\text{ext}}(\mathbf{r})$ can be written as a functional of a set of single-particle wave functions or Kohn-Sham orbitals $\{\psi_i(\mathbf{r})\}$

$$E[\{\psi_i(\mathbf{r})\}] = -\frac{1}{2} \int \psi_i(\mathbf{r}) \nabla^2 \psi_i(\mathbf{r}) d\mathbf{r} + \int \rho(\mathbf{r}) V_{\text{ext}}(\mathbf{r}) d\mathbf{r} + \frac{1}{2} \int \frac{\rho(\mathbf{r})\rho(\mathbf{r}')}{|\mathbf{r} - \mathbf{r}'|} d\mathbf{r} d\mathbf{r}' + E_{\text{xc}}[\rho(\mathbf{r})] \quad (1)$$

where single-particle electron density $\rho(\mathbf{r})$ is calculated as

$$\rho(\mathbf{r}) = \sum_i f_i \psi_i^*(\mathbf{r}) \psi_i(\mathbf{r}) \quad (2)$$

In Equation 2 the summation is done over all occupied single electronic states i and f_i is the occupation number of the i -th orbital. For doubly-occupied orbitals we have $f_i = 2$. In the usual setting in material science and chemistry, the external potential is usually the potential due to the atomic nuclei (ions):

$$V_{\text{ext}}(\mathbf{r}) = \sum_I \frac{Z_I}{|\mathbf{r} - \mathbf{R}_I|} \quad (3)$$

and an additional term of energy, the nucleus-nucleus energy E_{nn} , is added to the total energy functional (1):

$$E_{\text{nn}} = \frac{1}{2} \sum_I \sum_J \frac{Z_I Z_J}{|\mathbf{R}_I - \mathbf{R}_J|} \quad (4)$$

The energy terms in total energy functionals are the kinetic, external, Hartree, and exchange-correlation (XC) energy, respectively. The functional form of the last term (i.e. the XC energy) in terms of electron density is not known and one must resort to an approximation. In this article we will use the local density approximation for the XC energy. Under this approximation, the XC energy can be written as:

$$E_{\text{xc}}[\rho(\mathbf{r})] = \int \rho(\mathbf{r}) \epsilon_{\text{xc}}[\rho(\mathbf{r})] d\mathbf{r} \quad (5)$$

where ϵ_{xc} is the exchange-correlation energy per particle. There are many functional forms that have been devised for ϵ_{xc} and they are usually named by the persons who proposed them. An explicit form of ϵ_{xc} will be given later.

Many material properties can be derived from the minimum of the functional (1). This minimum energy is also called the *ground state energy*. This energy can be obtained by using direct minimization of the Kohn-Sham energy functional or by solving the Kohn-Sham equations:

$$\hat{H}_{\text{KS}} \psi_i(\mathbf{r}) = \epsilon_i \psi_i(\mathbf{r}) \quad (6)$$

where ϵ_i are the Kohn-Sham orbital energies and the Kohn-Sham Hamiltonian \hat{H}_{KS} is defined as

$$\hat{H}_{\text{KS}} = -\frac{1}{2} \nabla^2 + V_{\text{ext}}(\mathbf{r}) + V_{\text{Ha}}(\mathbf{r}) + V_{\text{xc}}(\mathbf{r}) \quad (7)$$

From the solutions of the Kohn-Sham equations: ϵ_i and $\psi_i(\mathbf{r})$ we can calculate the corresponding minimum total energy from the functional (1).

In the definition of Kohn-Sham Hamiltonian, other than the external potential which is usually specified from the problem, we have two additional potential terms, namely the Hartree and exchange-correlation potential. The Hartree potential can be calculated from its integral form

$$V_{\text{Ha}}(\mathbf{r}) = \int \frac{\rho(\mathbf{r}')}{|\mathbf{r} - \mathbf{r}'|} d\mathbf{r}' \quad (8)$$

An alternative way to calculate the Hartree potential is to solve the Poisson equation:

$$\nabla^2 V_{\text{Ha}}(\mathbf{r}) = -4\pi\rho(\mathbf{r}) \quad (9)$$

The exchange-correlation potential is defined as functional derivative of the exchange-correlation energy:

$$V_{\text{xc}}(\mathbf{r}) = \frac{\delta E[\rho(\mathbf{r})]}{\delta \rho(\mathbf{r})} \quad (10)$$

The Kohn-Sham equations are nonlinear eigenvalue equations in the sense that to calculate the solutions $\{\epsilon_i\}$ and $\{\psi_i(\mathbf{r})\}$ we need to know the electron density $\rho(\mathbf{r})$ to build the Hamiltonian. The electron density itself must be calculated from $\{\psi_i(\mathbf{r})\}$ which are not known (the quantities we want to solve for). The usual algorithm to solve the Kohn-Sham equations is the following:

- **(STEP 1)**: start from a guess input density $\rho^{\text{in}}(\mathbf{r})$
- **(STEP 2)**: calculate the Hamiltonian defined in (7)
- **(STEP 3)**: solve the eigenvalue equations in (6) to obtain the eigenpairs $\{\epsilon_i\}$, $\{\psi_i(\mathbf{r})\}$
- **(STEP 4)**: calculate the output electron density $\rho^{\text{out}}(\mathbf{r})$ from $\{\psi_i(\mathbf{r})\}$ that are obtained from the previous step.
- **(STEP 5)** Check whether the difference between $\rho^{\text{in}}(\mathbf{r})$ and $\rho^{\text{out}}(\mathbf{r})$ is small. If the difference is still above a chosen threshold then back to **STEP 1**. If the difference is already small the stop the algorithm.

The algorithm we have just described is known as the self-consistent field (SCF) algorithm.

Nowadays, there are many available software packages that can be used to solve the Kohn-Sham equations such as Quantum ESPRESSO, ABINIT, VASP, Gaussian, and NWChem are among the popular ones. A more complete list is given in a Wikipedia page [7]. These packages differs in several aspects, particularly the basis set used to discretize the Kohn-Sham equations and characteristic of the systems they can handle (periodic or non-periodic systems). These packages provide an easy way for researchers to carry out calculations based on density functional theory for particular systems they are interested in without knowing the details of how these calculations are performed.

In this article we will describe a simple way to solve the Kohn-Sham equations based on finite difference approximation. Our focus will be on the practical numerical implementation of a solver for the Kohn-Sham equations. The solver will be implemented using Julia programming language [8]. Our target is to calculate the ground state energy of several simple model systems. We will begin to build our solver starting from the ground up. The roadmap of the article is as follows.

- We begin from discussing numerical solution of Schroedinger equation in 1d. We will introduce finite difference approximation and its use in approximating second derivative operator that is present in the Schroedinger equation. We show how one can build the Hamiltonian matrix and solve the resulting eigenvalue equations using standard function that is available in Julia.
- In the next section, we discuss numerical solution of Schroedinger in 2d. We will introduce how one can handle 2d grid and applying finite difference approximation to the Laplacian operator present in the 2d Schroedinger equation. We also present several iterative methods to solve the eigenvalue equations.
- The next section discusses the numerical solution of Schroedinger equation in 3d. The methods presented in this section is a straightforward extension from the 2d case. In this section we start considering V_{ext} that is originated from the Coulomb interaction between atomic nuclei and electrons. We also introduce the concept of pseudopotential which is useful in practical calculations.
- The next section discusses about Poisson equation. Conjugate gradient method for solving system of linear equations. Hartree energy calculation.
- Hartree approximation, implementation of SCF algorithm
- Kohn-Sham equations, XC energy and potential, hydrogen molecule

2 Schroedinger equation in 1d

We are interested in finding the bound states of 1d time-independent Schroedinger equation:

$$\left[-\frac{1}{2} \frac{d^2}{dx^2} + V(x) \right] \psi(x) = E \psi(x) \quad (11)$$

with the boundary conditions:

$$\lim_{x \rightarrow \pm\infty} \psi(x) = 0 \quad (12)$$

This boundary condition is relevant for non-periodic systems such as atoms and molecules.

2.1 Grid points

First we need to define a spatial domain $[x_{\min}, x_{\max}]$ where x_{\min}, x_{\max} chosen such that the boundary condition 12 is approximately satisfied. The next step is to divide the spatial domain x using equally-spaced grid points which we will denote as $\{x_1, x_2, \dots, x_N\}$ where N is number of grid points. Various spatial quantities such as wave functions and potentials will be discretized on these grid points. The grid points $x_i, i = 1, 2, \dots$ are chosen as:

$$x_i = x_{\min} + (i - 1)h \quad (13)$$

where h is the spacing between the grid points:

$$h = \frac{x_{\max} - x_{\min}}{N - 1} \quad (14)$$

The following Julia code can be used to initialize the grid points:

```
function init_FD1d_grid( x_min::Float64, x_max::Float64, N::Int64 )
    L = x_max - x_min
    h = L / (N-1) # spacing
    x = zeros(Float64, N) # the grid points
    for i = 1:N
        x[i] = x_min + (i-1)*h
    end
    return x, h
end
```

2.2 Approximating second derivative

Our next task is to find an approximation to the second derivative operator present in the Equation (11). One simple approximation that we can use is the 3-point (central) finite difference:

$$\frac{d^2}{dx^2} \psi_i \approx \frac{\psi_{i+1} - 2\psi_i + \psi_{i-1}}{h^2} \quad (15)$$

where we have the following notation have been used: $\psi_i = \psi(x_i)$. By taking $\{\psi_i\}$ as a column vector, the second derivative operation can be expressed as matrix multiplication:

$$\vec{\psi}'' = \mathbb{D}^{(2)} \vec{\psi} \quad (16)$$

where $\mathbb{D}^{(2)}$ is the second derivative matrix operator:

$$\mathbb{D}^{(2)} = \frac{1}{h^2} \begin{bmatrix} -2 & 1 & 0 & 0 & 0 & \dots & 0 \\ 1 & -2 & 1 & 0 & 0 & \dots & 0 \\ 0 & 1 & -2 & 1 & 0 & \dots & 0 \\ \vdots & \ddots & \ddots & \ddots & \ddots & \ddots & \vdots \\ 0 & \dots & 0 & 1 & -2 & 1 & 0 \\ 0 & \dots & \dots & 0 & 1 & -2 & 1 \\ 0 & \dots & \dots & \dots & 0 & 1 & -2 \end{bmatrix} \quad (17)$$

An example implementation can be found in the following function.

```
function build_D2_matrix_3pt( N::Int64, h::Float64 )
    mat = zeros(Float64, N, N)
    for i = 1:N-1
        mat[i, i] = -2.0
        mat[i, i+1] = 1.0
        mat[i+1, i] = mat[i, i+1]
    end
    mat[N, N] = -2.0
    return mat/h^2
end
```

Before use this function to solve Schroedinger equation we will to test the operation in Equation (17) for a simple function which second derivative can be calculated analytically.

$$\psi(x) = e^{-\alpha x^2} \quad (18)$$

which second derivative can be calculated as

$$\psi''(x) = (-2\alpha + 4\alpha^2 x^2) e^{-\alpha x^2} \quad (19)$$

They are implemented in the following code

```

function my_gaussian(x; α=1.0)
    return exp(-α*x^2)
end

function d2_my_gaussian(x; α=1.0)
    return (-2*α + 4*α^2 * x^2) * exp(-α*x^2)
end

```




Figure 1: Finite difference approximation to a Gaussian function and its second derivative

In Figure 1, the comparison of analytical and numerical second derivative of a Gaussian function is shown. It can be seen clearly that the accuracy of numerical second derivative became better as the number of the grid points is increased.

2.3 Harmonic potential

Now that we know how to represent second derivative as a matrix, we are ready to solve the Schroedinger equation. We will start with a simple potential $V(x)$ for which we know the exact solutions, i.e. the harmonic potential:

$$V(x) = \frac{1}{2}\omega^2 x^2 \quad (20)$$

where ω is a parameter.

The Hamiltonian in the finite difference representation takes the following form:

$$\mathbb{H} = -\frac{1}{2}\mathbb{D}^{(2)} + \mathbb{V} \quad (21)$$

where \mathbb{V} is a diagonal matrix whose elements are:

$$\mathbb{V}_{ij} = V(x_i)\delta_{ij} \quad (22)$$

and $\mathbb{D}^{(2)}$ is the second derivative matrix defined previously.

The following code calculates the harmonic potential with the default value of $\omega = 1$.

```

function pot_harmonic(x; ω=1.0)
    return 0.5 * ω^2 * x^2
end

```

The following Julia snippet illustrates the steps of constructing the Hamiltonian matrix, starting from initialization of grid points, building the 2nd derivative matrix, and building the potential.

```

# Initialize the grid points
xmin = -5.0; xmax = 5.0
N = 51
x, h = init_FD1d_grid(xmin, xmax, N)
# Build 2nd derivative matrix
D2 = build_D2_matrix_3pt(N, h)
# Potential
Vpot = pot_harmonic.(x)

```

```
# Hamiltonian
Ham = -0.5*D2 + diag( 0 => Vpot )
```

Once the Hamiltonian matrix has been constructed, we can find the solutions or the eigenvalues and eigenvectors by solving the eigenproblem. In Julia, we can do this by calling the `eigen` function of `LinearAlgebra` package which is part of the standard Julia library. The following snippets shows how this can be achieved.

```
# Solve the eigenproblem
evals, evects = eigen( Ham )
# We will show the 5 lowest eigenvalues
Nstates = 5
@printf("Eigenvalues\n")
for i in 1:Nstates
    @printf("%5d %18.10f\n", i, evals[i])
end
```

We can compare our eigenvalues result with the analytical solution:

$$E_n = (2n - 1) \frac{\hbar}{2} \omega, \quad n = 1, 2, 3, \dots \quad (23)$$

The results are shown in the following table for $N = 51$.

n	Numerical	Exact	abs(error)
1	0.4987468513	0.5000000000	1.2531486828e-03
2	1.4937215179	1.5000000000	6.2784821079e-03
3	2.4836386480	2.5000000000	1.6361352013e-02
4	3.4684589732	3.5000000000	3.1541026791e-02
5	4.4481438504	4.5000000000	5.1856149551e-02

You may try to vary the number of N to achieve higher accuracy.




Figure 2: Eigenstates of harmonic oscillator

In addition to the eigenvalues, we also can visualize the eigenfunctions or eigenstates. The results are shown in Figure 2 for $N = 51$.

The full Julia program for this harmonic potential is given in `sch_1d/main_harmonic_01.jl`.

Note that calling `eigen` will give us N -pairs of eigenvalue-eigenfunctions where N is the dimension of the Hamiltonian matrix or in this case the number of grid points. We rarely needs all of these eigenpairs.

2.4 Higher order finite difference

To achieve more accurate result we can include more points in our calculations. However there is an alternative, namely by using more points or higher order formula to approximate second derivative. An example is 5-point formula for central difference approximation to second derivative:

$$\frac{d^2}{dx^2}\psi_i \approx \frac{-\psi_{i+2} + 16\psi_{i+1} - 30\psi_i + 16\psi_{i-1} - \psi_{i-2}}{12h^2} \quad (24)$$

The finite difference coefficients can be found in the literature or various sources on the web (for example: <http://web.media.mit.edu/~crtaylor/calculator.html>).

We have provided Julia codes for calculating second derivative matrix using 5, 7, and 9 points with the name `build_D2_matrix_xpt.jl` where $x = 5, 7, 9$. You can repeat the calculation for harmonic oscillator potential with fixed number of N and compare the eigenvalue results by using 3, 5, 7, and 9 points formula for second derivative matrix.

3 Schroedinger equation in 2d

Now we will turn our attention to higher dimensions, i.e 2d. The Schrodinger equation in 2d reads:

$$\left[-\frac{1}{2}\nabla^2 + V(x, y) \right] \psi(x, y) = E \psi(x, y) \quad (25)$$

where ∇^2 is the Laplacian operator:

$$\nabla^2 = \frac{\partial^2}{\partial x^2} + \frac{\partial^2}{\partial y^2} \quad (26)$$

3.1 Describing grid in 2d

Now we have two directions x and y . Our approach to solving the Schroedinger equation is similar to the one we have used before in 1d, however several technical difficulties will arise.

To describe the computational grid, we now need to specify x_{\max}, x_{\min} for the X-domain and y_{\max}, y_{\min} for Y-domain. We also need to specify number of grid points in for each x and y -directions, i.e. N_x and N_y . That are quite lot of variables. For easier management, we will collect our grid related variables in one data structure or `struct` in Julia. A `struct` in Julia looks very much like C-struct. It also defines a new custom data type in Julia.

Our `struct` definition looks like this.

```
struct FD2dGrid
    Npoints::Int64
    Nx::Int64
    Ny::Int64
    hx::Float64
    hy::Float64
    dA::Float64
    x::Array{Float64,1}
    y::Array{Float64,1}
    r::Array{Float64,2}
    idx_ip2xy::Array{Int64,2}
    idx_xy2ip::Array{Int64,2}
end
```

An instance of `FD2dGrid` can be initialized using the following constructor function:

```
function FD2dGrid( x_domain, Nx, y_domain, Ny )
    x, hx = init_FD1d_grid(x_domain, Nx)
    y, hy = init_FD1d_grid(y_domain, Ny)
    dA = hx*hy
    Npoints = Nx*Ny
    r = zeros(2,Npoints)
    ip = 0
    idx_ip2xy = zeros(Int64,2,Npoints)
    idx_xy2ip = zeros(Int64,Nx,Ny)
    for j in 1:Ny
        for i in 1:Nx
            ip = ip + 1
            r[1,ip] = x[i]
```

```

        r[2,ip] = y[j]
        idx_ip2xy[1,ip] = i
        idx_ip2xy[2,ip] = j
        idx_xy2ip[i,j] = ip
    end
end
return FD2dGrid(Npoints, Nx, Ny, hx, hy, dA, x, y, r, idx_ip2xy, idx_xy2ip)
end

```

A short explanation about the members of FD2dGrid follows.

- `Npoints` is the total number of grid points.
- `Nx` and `Ny` is the total number of grid points in x and y -directions, respectively.
- `hx` and `hy` is grid spacing in x and y -directions, respectively. `dA` is the product of `hx` and `hy`.
- `x` and `y` are the grid points in x and y -directions. The actual two dimensional grid points $r \equiv (x_i, y_i)$ are stored as two dimensional array `r`.
- The two integers arrays `idx_ip2xy` and `idx_xy2ip` defines mapping between two dimensional grids and linear grids.

As an illustration let's build a grid for a rectangular domain $x_{\min} = y_{\min} = -5$ and $x_{\max} = y_{\max} = 5$ and $N_x = 3$, $N_y = 4$. Using the above constructor for FD2dGrid:

```

Nx = 3
Ny = 4
fdgrid = FD2dGrid( (-5.0,5.0), Nx, (-5.0,5.0), Ny )

```

Dividing the x and y accordingly we obtain $N_x = 3$ grid points along x -direction

```

> println(fdgrid.x)
[-5.0, 0.0, 5.0]

```

and $N_y = 4$ points along the y -direction

```

> println(fdgrid.y)
[-5.0, -1.6666666666666665, 1.6666666666666667, 5.0]

```

The actual grid points are stored in `fdgrid.r`. Using the following snippet, we can printout all of the grid points:

```

for ip = 1:fdgrid.Npoints
    @printf("%3d %8.3f %8.3f\n", ip, fdgrid.r[1,ip], fdgrid.r[2,ip])
end

```

The results are:

```

 1  -5.000  -5.000
 2   0.000  -5.000
 3   5.000  -5.000
 4  -5.000  -1.667
 5   0.000  -1.667
 6   5.000  -1.667
 7  -5.000   1.667
 8   0.000   1.667
 9   5.000   1.667
10  -5.000   5.000
11   0.000   5.000
12   5.000   5.000

```

We also can use the usual rearrange these points in the usual 2d grid rearrangement:

```

[ -5.000, -5.000] [ -5.000, -1.667] [ -5.000, 1.667] [ -5.000, 5.000]
[  0.000, -5.000] [  0.000, -1.667] [  0.000, 1.667] [  0.000, 5.000]
[  5.000, -5.000] [  5.000, -1.667] [  5.000, 1.667] [  5.000, 5.000]

```

which can be produced from the following snippet:

```
for i = 1:Nx
    for j = 1:Ny
        ip = fdgrid.idx_xy2ip[i,j]
        @printf("[%8.3f,%8.3f] ", fdgrid.r[1,ip], fdgrid.r[2,ip])
    end
    @printf("\n")
end
```

3.2 Laplacian operator

Having built out 2d grid, we now turn our attention to the second derivative operator or the Laplacian in the equation 25. There are several ways to build a matrix representation of the Laplacian, but we will use the easiest one.

Before constructing the Laplacian matrix, there is an important observation that we should make about the second derivative matrix $\mathbb{D}^{(2)}$. We should note that the second derivative matrix contains mostly zeros. This type of matrix that most of its elements are zeros is called **sparse matrix**. In a sparse matrix data structure, we only store its non-zero elements with specific formats such as compressed sparse row/column format (CSR/CSC) and coordinate format. We have not made use of the sparsity of the second derivative matrix in the 1d case for simplicity. In the higher dimensions, however, we must make use of this sparsity, otherwise we will waste computational resources by storing many zeros. The Laplacian matrix that we will build from $\mathbb{D}^{(2)}$ is also very sparse.

Given second derivative matrix in x , $\mathbb{D}_x^{(2)}$, y direction, $\mathbb{D}_y^{(2)}$, we can construct finite difference representation of the Laplacian operator \mathbb{L} by using

$$\mathbb{L} = \mathbb{D}_x^{(2)} \otimes \mathbb{I}_y + \mathbb{I}_x \otimes \mathbb{D}_y^{(2)} \quad (27)$$

where \otimes is Kronecker product. In Julia, we can use the function `kron` to form the Kronecker product between two matrices A and B as `kron(A,B)`.

The following function illustrates the above approach to construct matrix representation of the Laplacian operator.

```
function build_nabla2_matrix( fdgrid::FD2dGrid; func_1d=build_D2_matrix_3pt )
    Nx = fdgrid.Nx
    hx = fdgrid.hx
    Ny = fdgrid.Ny
    hy = fdgrid.hy

    D2x = func_1d(Nx, hx)
    D2y = func_1d(Ny, hy)

    ∇² = kron(D2x, speye(Ny)) + kron(speye(Nx), D2y)
    return ∇²
end
```

In the Figure 3, an example to the approximation of 2nd derivative of 2d Gaussian function by using finite difference is shown.




Figure 3: Two-dimensional Gaussian function and its finite difference approximation of second derivative

3.3 Iterative methods for eigenvalue problem

Now that we know how to build the Laplacian matrix, we now can build the Hamiltonian matrix given some potential:

```
∇² = build_nabla2_matrix( fdgrid )
Ham = -0.5*∇² + spdiags( 0 => Vpot )
```

Note that we have used sparse diagonal matrix for building the potential matrix by using the function `spdiags`. Our next task after building the Hamiltonian matrix is to find the eigenvalues and eigenfunctions. However, note that the Hamiltonian matrix size is large. For example, if we use $N_x = 50$ and $N_y = 50$ we will end up with a Hamiltonian matrix with the size of 2500. The use `eigen` method to solve this eigenvalue problem is thus not practical. Actually, given enough computer memory and time, we can use the function `eigen` anyway to find the eigenvalue and eigenfunction of the Hamiltonian, however it is not recommended nor practical for larger problem size. Typically, we also do not need to solve for all eigenvalue and eigenfunction pairs. We only need to solve for several eigenpairs with lowest eigenvalues. In typical density functional theory calculations, we typically solve for $N_{\text{electrons}}$ or $N_{\text{electrons}}/2$ lowest states, where $N_{\text{electrons}}$ is the number of electrons in the system.

In numerical methods, there are several methods to search for several eigenpairs of a matrix. These methods falls into the category of *partial or iterative diagonalization methods*. Several known methods are Lanczos method, Davidson method, preconditioned conjugate gradients, etc.

In this short article, we will not discuss about these methods in depth. However, we have prepared several implementation of iterative diagonalization methods for your convenience:

- `diag_Emin_PCG`
- `diag_davidson`
- `diag_LOBPCG`

Almost all iterative methods need a good preconditioner to function properly. In this talk, we will use several preconditioners that have been implemented in several packages in Julia such as incomplete LU and multigrid preconditioners. Here we show an example of Julia program to solve the Schroedinger equation for two dimensional harmonic potentials. The complete program can be found in the `sch_2d/main_harmonic.jl`.

```
function pot_harmonic( fdgrid::FD2dGrid; ω=1.0 )
    Npoints = fdgrid.Npoints
    Vpot = zeros(Npoints)
    for i in 1:Npoints
        x = fdgrid.r[1,i]
        y = fdgrid.r[2,i]
        Vpot[i] = 0.5 * ω^2 * ( x^2 + y^2 )
    end
    return Vpot
end

function main()
    Nx = 50
    Ny = 50
    fdgrid = FD2dGrid( (-5.0,5.0), Nx, (-5.0,5.0), Ny )
    V2 = build_nabla2_matrix( fdgrid )
    Vpot = pot_harmonic( fdgrid )
    Ham = -0.5*V2 + spdiags( 0 => Vpot )

    # Preconditioner based on inverse kinetic
    prec = ilu(-0.5*V2)


    Nstates = 10
    Npoints = Nx*Ny
    X = rand(Float64, Npoints, Nstates)
    ortho_sqrt!(X)

    evals = diag_LOBPCG!( Ham, X, prec, verbose=true )
    X = X/sqrt(fdgrid.dA) # renormalize the eigenfunctions

    @printf("\n\nEigenvalues\n")
    for i in 1:Nstates
        @printf("%5d %18.10f\n", i, evals[i])
    end
end
```

end

The eigenfunctions are shown in Figure 3.3.



Eigenvalues ($N_x = N_y = 50$):

1	0.9999999862
2	1.9999998768
3	1.9999999392
4	2.9999992436
5	2.9999993054
6	2.9999997845
7	3.9999973668
8	3.9999976649
9	3.9999992565
10	3.9999998030

Energy

$n_x + n_y + 1$	&	Values of n_x and n_y
1	&	(0, 0)
2	&	(1, 0) (0, 1)
3	&	(2, 0) (1, 1) (1, 1)
4	&	(3, 0) (0, 3) (2, 1) (1, 2)

Energy:

$$E_{n_x+n_y} = \hbar\omega (n_x + n_y + 1) \quad (28)$$

4 Schroedinger equation in 3d

The 3d case of Schroedinger equation is a straightforward extension of the 2d case. The Schroedinger equation thus reads:

$$\left[-\frac{1}{2}\nabla^2 + V(x, y, z) \right] \psi(x, y, z) = E \psi(x, y, z) \quad (29)$$

where ∇^2 is the Laplacian operator:

$$\nabla^2 = \frac{\partial^2}{\partial x^2} + \frac{\partial^2}{\partial y^2} + \frac{\partial^2}{\partial z^2} \quad (30)$$

We begin by defining a struct called `FD3dGrid` which is a straightforward generalization of `FD2dGrid`. The implementation of this struct can be found in the file `FD3d/FD3dGrid.jl`.

The Lagrangian operator in 3d also can be implemented by straightforward extension of 2d case.

```
const ⊗ = kron

function build_nabla2_matrix( fdgrid::FD3dGrid; func_1d=build_D2_matrix_3pt )

    D2x = func_1d(fdgrid.Nx, fdgrid.hx)
    D2y = func_1d(fdgrid.Ny, fdgrid.hy)
    D2z = func_1d(fdgrid.Nz, fdgrid.hz)

    IIx = speye(fdgrid.Nx)
    IIy = speye(fdgrid.Ny)
    IIz = speye(fdgrid.Nz)

    ∇² = D2x⊗IIy⊗IIz + IIx⊗D2y⊗IIz + IIx⊗IIy⊗D2z

    return ∇²
end
```

The main difference is that we have used the symbol \otimes in place of `kron` function to make our code simpler. We hope that at this point you will have no difficulties to create your own 3d Schroedinger equation solver. Analytic solution for energy:

$$E_{n_x+n_y+n_z} = \hbar\omega \left(n_x + n_y + n_z + \frac{3}{2} \right) \quad (31)$$

Degeneracies:

$$g_n = \frac{(n+1)(n+2)}{2} \quad (32)$$

```
n = n_x + n_y + n_z

n = 0: (1) (2) / 2 = 1
n = 1: (2) (3) / 2 = 3
n = 2: (3) (4) / 2 = 6
```

4.1 Hydrogen atom and an introduction to pseudopotential

Until now, we only have considered simple potentials such as harmonic potential. Now we will move on and consider more realistic potentials which is used in practical electronic calculations.

For most applications in materials physics and chemistry the external potential that is felt by electrons is the Coulombic potential due to atomic nucleus. This potential has the following form:

$$V(r) = - \sum_I^{N_{\text{atom}}} \frac{Z_I}{|\mathbf{r} - \mathbf{R}_I|} \quad (33)$$

where R_I are the positions and Z_I are the charges of the atomic nucleus present in the system. We will consider the most simplest system, namely the hydrogen atom $Z_I = 1$, for which we have

$$V(r) = - \frac{1}{|\mathbf{r} - \mathbf{R}_0|} \quad (34)$$

The following Julia code implement the H atom potential:

```
function pot_H_atom( fdgrid::FD3dGrid; r0=(0.0, 0.0, 0.0) )
    Npoints = fdgrid.Npoints
    Vpot = zeros(Npoints)
    for i in 1:Npoints
        dx = fdgrid.r[1,i] - r0[1]
        dy = fdgrid.r[2,i] - r0[2]
        dz = fdgrid.r[3,i] - r0[3]
        Vpot[i] = -1.0/sqrt(dx^2 + dy^2 + dz^2)
    end
end
```

```

end
return Vpot
end

```

With only minor modification to our program for harmonic potential, we can solve the Schroedinger equation for the hydrogen atom:

```

fdgrid = FD3dGrid( (-5.0,5.0), Nx, (-5.0,5.0), Ny, (-5.0,5.0), Nz )
V2 = build_nabla2_matrix( fdgrid, func_1d=build_D2_matrix_9pt )
Vpot = pot_H_atom( fdgrid )
Ham = -0.5*V2 + spdiagm( 0 => Vpot )
prec = aspreconditioner(ruge_stuben(Ham))
Nstates = 1 # only choose the lowest lying state
Npoints = Nx*Ny*Nz
X = ortho_sqrt( rand(Float64, Npoints, Nstates) ) # random initial guess of wave function
evals = diag_LOBPCG!( Ham, X, prec, verbose=true )

```

For the grid size of $N_x = N_y = N_z = 50$ and using 9-point finite-difference approximation to the second derivative operator in 1d we obtain the eigenvalue of -0.4900670759 Ha which is not too bad if compared with the exact value of -0.5 Ha. We can try to increase the grid size until we can get satisfactory result.


Note that there is a caveat when we are trying to use the Coulombic potential. This potential is divergent at $r = 0$, so care must be taken such that this divergence is not encountered in our potential. We have tried to achieve this by using choosing the numbers N_x , N_y , and N_z to be even numbers. This way, we have avoiding encountering divergence in the calculation of Coulomb potential.

In the many electronic structure calculations, it is sometime convenient to replace the Coulomb potential with another potential which is smoother which we will refer to as a **pseudopotential**. Not all smooth potentials can do the job. The smooth potential should satisfy several requirements. One of the most important requirement is that the smooth potential should have similar scattering properties as the original Coulomb potential that it replaces. This means that the potential should have posses similar eigenvalues as the original Coulomb potential. Usually we don't try to reproduce all eigenvalue spectrum but only the eigenvalues which belongs to the valence electrons. The valence electrons are responsible for most chemically and physically important properties so this is an acceptable approximation for most cases.

The theory and algorithms for constructing pseudopotentials are beyond the scope of this article. Most pseudopotentials are non-local by construction and this can make our program rather complicated. In this article we focus on the so-called local pseudopotential. Practically, local pseudopotentials pose no additional difficulties as the potentials that we have considered so far. As an example of a pseudopotential, we will consider the following local pseudopotential for hydrogen atom:

$$V_{H,ps}(r) = -\frac{Z_{val}}{r} \operatorname{erf}\left(\frac{\bar{r}}{\sqrt{2}}\right) + \exp\left(-\frac{1}{2}\bar{r}^2\right) (C_1 + C_2\bar{r}^2) \quad (35)$$

where $\bar{r} = r/r_{loc}$ and with the parameters $r_{loc} = 0.2$, $Z_{val} = 1$, $C_1 = -4.0663326$, and $C_2 = 0.6678322$.



The code

```

function pot_Hps_HGH( fdgrid::FD3dGrid; r0=(0.0, 0.0, 0.0) )
    Npoints = fdgrid.Npoints
    Vpot = zeros( Float64, Npoints )

    # Parameters
    Zval = 1
    rloc = 0.2

```

```

C1 = -4.0663326
C2 = 0.6678322
for ip = 1:Npoints
    dx2 = ( fdgrid.r[1,ip] - r0[1] )^2
    dy2 = ( fdgrid.r[2,ip] - r0[2] )^2
    dz2 = ( fdgrid.r[3,ip] - r0[3] )^2
    r = sqrt(dx2 + dy2 + dz2)
    if r < eps()
        Vpot[ip] = -2*Zval/(sqrt(2*pi)*rloc) + C1
    else
        rrloc = r/rloc
        Vpot[ip] = -Zval/r * erf( r/(sqrt(2.0)*rloc) ) +
            (C1 + C2*rrloc^2)*exp(-0.5*(rrloc)^2)
    end
end
return Vpot
end

```

5 Poisson equation

In this section we will discuss a second equation that is important in solving Kohn-Sham equation, namely the Poisson equation. In the context of solving Kohn-Sham equation, Poisson equation is used to calculate classical electrostatic potential due to some electronic charge density. The Poisson equation that we will solve have the following form:

$$\nabla^2 V_{\text{Ha}}(\mathbf{r}) = -4\pi\rho(\mathbf{r}) \quad (36)$$

where $\rho(\mathbf{r})$ is the electronic density. Using finite difference discretization for the operator ∇^2 we end up with the following linear equation:

$$\mathbb{L}\mathbf{V} = \mathbf{f} \quad (37)$$

where \mathbb{L} is the matrix representation of the Laplacian operator \mathbf{f} is the discrete representation of the right hand side of the equation 36, and the unknown \mathbf{V} is the discrete representation of the Hartree potential.

There exist several methods for solving the linear equation 37. We will use the so-called conjugate gradient method for solving this equation. This method is an iterative method, so it generally needs a good preconditioner to achieve good convergence. A detailed derivation about the algorithm is beyond this article and the readers are referred to several existing literatures [9, 10] and a webpage [11] for more information. The algorithm is described in Poisson_solve_PCG.jl

```

function Poisson_solve_PCG( Lmat::SparseMatrixCSC{Float64,Int64},
    prec,
    f::Array{Float64,1}, NiterMax::Int64;
    TOL=5.e-10 )

    Npoints = size(f,1)
    phi = zeros( Float64, Npoints )
    r = zeros( Float64, Npoints )
    p = zeros( Float64, Npoints )
    z = zeros( Float64, Npoints )
    nabla2_phi = Lmat*phi
    r = f - nabla2_phi
    z = copy(r)
    ldiv!(prec, z)
    p = copy(z)
    rsold = dot( r, z )
    for iter = 1 : NiterMax
        nabla2_phi = Lmat*p
        alpha = rsold/dot( p, nabla2_phi )
        phi = phi + alpha * p
        r = r - alpha * nabla2_phi
        z = copy(r)
        ldiv!(prec, z)
        rsnew = dot(z, r)
        deltars = rsold - rsnew
        if sqrt(abs(rsnew)) < TOL
            break
        end
        p = z + (rsnew/rsold) * p
    end
end

```

```

        rsold = rsnew
    end
    return phi
end

```

To test our implementation we will adopt a problem given in Prof. Arias Practical DFT mini-course [12]. In this problem we will solve Poisson equation for a given charge density built from superposition of two Gaussian charge density:

$$\rho(\mathbf{r}) = \frac{1}{(2\pi\sigma_1^2)^{\frac{3}{2}}} \exp\left(-\frac{\mathbf{r}^2}{2\sigma_1^2}\right) - \frac{1}{(2\pi\sigma_2^2)^{\frac{3}{2}}} \exp\left(-\frac{\mathbf{r}^2}{2\sigma_2^2}\right) \quad (38)$$

After we obtain $V_{\text{Ha}}(\mathbf{r})$, we calculate the Hartree energy:

$$E_{\text{Ha}} = \frac{1}{2} \int \rho(\mathbf{r}) V_{\text{Ha}}(\mathbf{r}) d\mathbf{r} \quad (39)$$

and compare the result with the analytical formula.

```

function test_main( NN::Array{Int64} )
    AA = [0.0, 0.0, 0.0]
    BB = [16.0, 16.0, 16.0]
    # Initialize grid
    FD = FD3dGrid( NN, AA, BB )
    # Box dimensions
    Lx = BB[1] - AA[1]
    Ly = BB[2] - AA[2]
    Lz = BB[3] - AA[3]
    # Center of the box
    x0 = Lx/2.0
    y0 = Ly/2.0
    z0 = Lz/2.0
    # Parameters for two gaussian functions
    sigma1 = 0.75
    sigma2 = 0.50

    Npoints = FD.Nx * FD.Ny * FD.Nz
    rho = zeros(Float64, Npoints)
    phi = zeros(Float64, Npoints)
    # Initialization of charge density
    dr = zeros(Float64, 3)
    for ip in 1:Npoints
        dr[1] = FD.r[1,ip] - x0
        dr[2] = FD.r[2,ip] - y0
        dr[3] = FD.r[3,ip] - z0
        r = norm(dr)
        rho[ip] = exp( -r^2 / (2.0*sigma2^2) ) / (2.0*pi*sigma2^2)^1.5 -
                  exp( -r^2 / (2.0*sigma1^2) ) / (2.0*pi*sigma1^2)^1.5
    end
    deltaV = FD.hx * FD.hy * FD.hz
    Laplacian3d = build_nabla2_matrix( FD, func_1d=build_D2_matrix_9pt )
    prec = aspreconditioner(ruge_stuben(Laplacian3d))
    @printf("Test norm charge: %18.10f\n", sum(rho)*deltaV)
    print("Solving Poisson equation:\n")
    phi = Poisson_solve_PCG( Laplacian3d, prec, -4*pi*rho, 1000, verbose=true, TOL=1e-10 )
    # Calculation of Hartree energy
    Unum = 0.5*sum( rho .* phi ) * deltaV
    Uana = ((1.0/sigma1 + 1.0/sigma2)/2.0 - sqrt(2.0)/sqrt(sigma1^2 + sigma2^2))/sqrt(pi)
    @printf("Numeric = %18.10f\n", Unum)
    @printf("Uana = %18.10f\n", Uana)
    @printf("abs diff = %18.10e\n", abs(Unum-Uana))
end
test_main([64,64,64])

```

6 Hartree calculation

Ignoring the XC potential.

We will introduce self-consistent field (SCF) method.

New data structure: Hamiltonian

```

mutable struct Hamiltonian
    fdgrid::FD3dGrid
    Laplacian::SparseMatrixCSC{Float64,Int64}
    V_Ps_loc::Vector{Float64}
    V_Hartree::Vector{Float64}
    rhoe::Vector{Float64}
    precKin
    precLaplacian
end

```

```

function Hamiltonian( fdgrid::FD3dGrid, ps_loc_func::Function; func_1d=build_D2_matrix_5pt
↪ )

    Laplacian = build_nabla2_matrix( fdgrid, func_1d=func_1d )

    V_Ps_loc = ps_loc_func( fdgrid )
    Npoints = fdgrid.Npoints

    V_Hartree = zeros(Float64, Npoints)

    Rhoe = zeros(Float64, Npoints)

    precKin = aspreconditioner( ruge_stuben(-0.5*Laplacian) )
    precLaplacian = aspreconditioner( ruge_stuben(Laplacian) )

    return Hamiltonian( fdgrid, Laplacian, V_Ps_loc, V_Hartree, Rhoe, precKin,
↪ precLaplacian )
end

```

```

import Base: *
function *( Ham::Hamiltonian, psi::Matrix{Float64} )
    Nbasis = size(psi,1)
    Nstates = size(psi,2)
    Hpsi = zeros(Float64,Nbasis,Nstates)
    # include occupation number factor
    Hpsi = -0.5*Ham.Laplacian * psi ## 2.0
    for ist in 1:Nstates, ip in 1:Nbasis
        Hpsi[ip,ist] = Hpsi[ip,ist] + ( Ham.V_Ps_loc[ip] + Ham.V_Hartree[ip] ) *
↪ psi[ip,ist]
    end
    return Hpsi
end

```

```

function update!( Ham::Hamiltonian, Rhoe::Vector{Float64} )
    Ham.rhoe[:] = Rhoe[:]
    Ham.V_Hartree = Poisson_solve_PCG( Ham.Laplacian, Ham.precLaplacian, -4*pi*Rhoe, 1000,
↪ verbose=false, TOL=1e-10 )
    return
end

```

Calculate electron density:

```

function calc_rhoe( psi::Array{Float64,2} )
    Nbasis = size(psi,1)
    Nstates = size(psi,2)
    Rhoe = zeros(Float64,Nbasis)
    for ist in 1:Nstates
        for ip in 1:Nbasis
            Rhoe[ip] = Rhoe[ip] + 2.0*psi[ip,ist]*psi[ip,ist]
        end
    end
    return Rhoe
end

```

Calculate energy terms:

```

mutable struct Energies
    Kinetic::Float64
    Ps_loc::Float64
    Hartree::Float64
end

import Base: sum
function sum( ene::Energies )
    return ene.Kinetic + ene.Ps_loc + ene.Hartree
end

function calc_E_kin( Ham, psi::Array{Float64,2} )
    Nbasis = size(psi,1)
    Nstates = size(psi,2)
    E_kin = 0.0
    nabla2psi = zeros(Float64,Nbasis)
    dVol = Ham.fdggrid.dVol
    # Assumption: Focc = 2 for all states
    for ist in 1:Nstates
        @views nabla2psi = -0.5*Ham.Laplacian*psi[:,ist]
        E_kin = E_kin + 2.0*dot( psi[:,ist], nabla2psi[:] )*dVol
    end
    return E_kin
end

function calc_energies( Ham::Hamiltonian, psi::Array{Float64,2} )
    dVol = Ham.fdggrid.dVol
    E_kin = calc_E_kin( Ham, psi )
    E_Ps_loc = sum( Ham.V_Ps_loc .* Ham.rhoe )*dVol
    E_Hartree = 0.5*sum( Ham.V_Hartree .* Ham.rhoe )*dVol

    return Energies(E_kin, E_Ps_loc, E_Hartree)
end

```

Self-consistent field:

```

function pot_harmonic( fdggrid::FD3dGrid; w=1.0, center=[0.0, 0.0, 0.0] )
    Npoints = fdggrid.Npoints
    Vpot = zeros(Npoints)
    for i in 1:Npoints
        x = fdggrid.r[1,i] - center[1]
        y = fdggrid.r[2,i] - center[2]
        z = fdggrid.r[3,i] - center[3]
        Vpot[i] = 0.5 * w^2 * ( x^2 + y^2 + z^2 )
    end
    return Vpot
end

function main()
    AA = [-3.0, -3.0, -3.0]
    BB = [3.0, 3.0, 3.0]
    NN = [25, 25, 25]

    fdggrid = FD3dGrid( NN, AA, BB )
    my_pot_harmonic( fdggrid ) = pot_harmonic( fdggrid, w=2 )
    Ham = Hamiltonian( fdggrid, my_pot_harmonic, func_1d=build_D2_matrix_9pt )

    Nbasis = prod(NN)
    dVol = fdggrid.dVol
    Nstates = 4
    psi = rand(Float64,Nbasis,Nstates)
    ortho_sqrt!(psi)
    psi = psi/sqrt(dVol)

    Rhoe = calc_rhoe( psi )
    @printf("Integrated Rhoe = %18.10f\n", sum(Rhoe)*dVol)
    update!( Ham, Rhoe )
end

```



```

evals = zeros(Float64,Nstates)
Etot_old = 0.0
dEtot = 0.0
betamix = 0.5
dRhoe = 0.0
NiterMax = 100

for iterSCF in 1:NiterMax
    evals = diag_LOBPCG!( Ham, psi, Ham.precKin, verbose_last=true )
    psi = psi/sqrt(dVol)
    Rhoe_new = calc_rhoe( psi )
    @printf("Integrated Rhoe_new = %18.10f\n", sum(Rhoe_new)*dVol)
    Rhoe = betamix*Rhoe_new + (1-betamix)*Rhoe
    @printf("Integrated Rhoe      = %18.10f\n", sum(Rhoe)*dVol)
    update!( Ham, Rhoe )
    Etot = sum( calc_energies( Ham, psi ) )
    dRhoe = norm(Rhoe - Rhoe_new)
    dEtot = abs(Etot - Etot_old)
    @printf("%5d %18.10f %18.10e %18.10e\n", iterSCF, Etot, dEtot, dRhoe)
    if dEtot < 1e-6
        @printf("Convergence is achieved in %d iterations\n", iterSCF)
        for i in 1:Nstates
            @printf("%3d %18.10f\n", i, evals[i])
        end
        break
    end

    Etot_old = Etot
end

```

7 Kohn-Sham calculations

Using XC

Introduction of module:

Electrons type:

```

mutable struct Electrons
    Nelectrons::Int64
    Nstates::Int64
    Nstates_occ::Int64
    Focc::Array{Float64,1}
    energies::Array{Float64,1}
end

```

```

function Electrons( Nelectrons::Int64; Nstates_extra=0 )
    is_odd = (Nelectrons%2 == 1)
    Nstates_occ = round(Int64, Nelectrons/2)
    if is_odd
        Nstates_occ = Nstates_occ + 1
    end
    Nstates = Nstates_occ + Nstates_extra
    Focc = zeros(Float64,Nstates)
    energies = zeros(Float64,Nstates)
    if !is_odd
        for i in 1:Nstates_occ
            Focc[i] = 2.0
        end
    else
        for i in 1:Nstates_occ-1
            Focc[i] = 2.0
        end
        Focc[Nstates_occ] = 1.0
    end

```

```

    return Electrons(Nelectrons, Nstates, Nstates_occ, Focc, energies)
end

```

Example use of Electrons:

New Hamiltonian:

```

mutable struct Hamiltonian
    fdgrid::FD3dGrid
    Laplacian::SparseMatrixCSC{Float64,Int64}
    V_Ps_loc::Vector{Float64}
    V_Hartree::Vector{Float64}
    V_XC::Vector{Float64}
    electrons::Electrons
    rhoe::Vector{Float64}
    precKin
    precLaplacian
    energies::Energies
end

```

Update the potential:

```

function update!( Ham::Hamiltonian, Rhoe::Vector{Float64} )
    Ham.rhoe = Rhoe
    Ham.V_Hartree = Poisson_solve_PCG( Ham.Laplacian, Ham.precLaplacian, -4*pi*Rhoe, 1000,
    ↪ verbose=false, TOL=1e-10 )
    Ham.V_XC = excVWN( Rhoe ) + Rhoe .* excpVWN( Rhoe )
    return
end

```

Application of Hamiltonian

```

import Base: *
function *( Ham::Hamiltonian, psi::Matrix{Float64} )
    Nbasis = size(psi,1)
    Nstates = size(psi,2)
    Hpsi = zeros(Float64,Nbasis,Nstates)
    Hpsi = -0.5*Ham.Laplacian * psi
    for ist in 1:Nstates, ip in 1:Nbasis
        Hpsi[ip,ist] = Hpsi[ip,ist] + ( Ham.V_Ps_loc[ip] + Ham.V_Hartree[ip] +
    ↪ Ham.V_XC[ip] ) * psi[ip,ist]
    end
    return Hpsi
end

```

Appendices

A Short introduction to Julia programming language

B Conjugate gradient for system of linear equations

C Iterative diagonalization method

References

- [1] P. Hohenberg and W. Kohn. Inhomogenous electron gas. *Phys. Rev.*, 136:B864–871, 1964.
- [2] W Kohn and L.J. Sham. Self-consistent equations including exchange and correlation effects. *Phys. Rev.*, 140(4A):A1333–1138, 1965.
- [3] T. van Mourik T, M Bühl, and M-P Gaigeot. Density functional theory across chemistry, physics and biology. *Phil. Trans. R. Soc. A*, 372:20120488, 2014.

- [4] R.M. Martin. *Electronic Structure, Basic Theory and Practical Methods*. Cambridge University Press, Cambridge, UK, 2004.
- [5] Jorge Kohanoff. *Electronic structure calculations for solids and molecules: theory and computational methods*. Cambridge University Press, 2006.
- [6] Dominik Marx and Jürg Hutter. *Ab Initio Molecular Dynamics: Basic Theory and Advanced Methods*. Cambridge University Press, 2009.
- [7] List of quantum chemistry and solid-state physics software. https://en.wikipedia.org/wiki/List_of_quantum_chemistry_and_solid-state_physics_software.
- [8] Julia programming language. <https://julialang.org/>.
- [9] Magnus R. Hestenes and Eduard Stiefel. Methods of conjugate gradients for solving linear systems. *Journal of Research of the National Bureau of Standards*, 49:409, 1952.
- [10] Jonathan Richard Shewchuk. An introduction to the conjugate gradient method without the agonizing pain. Available <https://www.cs.cmu.edu/~quake-papers/painless-conjugate-gradient.pdf>.
- [11] Conjugate gradient method. https://en.wikipedia.org/wiki/Conjugate_gradient_method.
- [12] Practical DFT mini-course. <http://jdftx.org/PracticalDFT.html>.